
Count between dates by age range

Summary 

To count values between two dates that also fall into specific numeric ranges, you can use a
formula based on the COUNTIFS function, with help from the LEFT, RIGHT, FIND, and LEN
functions. In the example shown the formula in H8, copied down, is:

= COUNTIFS(joined,">=" & start,joined,"<=" & end, 
age,">=" & LEFT(G8,FIND("-",G8) - 1), 
age,"<=" & RIGHT(G8,LEN(G8) - FIND("-",G8)))

where start (H4), end (H5), age (D5:D16), and joined (E5:E16) are named ranges. At each new
row, the formula returns the count of all rows between start and end dates (inclusive) that also fall
into the age range as shown in column G.

Explanation 

The goal of this example is to count rows in the data where the date joined falls between start and
end dates (inclusive) and the age also falls into the age ranges seen in column G. The formula is
complicated somewhat by the fact that the age range labels are actually text, so we need to extract
a low and high number for each age range as a separate step.

Note: the named ranges shown in this example are entirely optional. They are a way to make the
formula easier to enter, read, and copy.

Count between dates

The main function used to solve this problem is COUNTIFS. To explain how this works, we'll look
first with at total seen in cell H6. This total does not take into account age groups, it simply counts
all records that fall between the start and end dates. The formula in H6 is:

= COUNTIFS(joined,">=" & start,joined,"<=" & end)

COUNTIFS is configured with two range/criteria pairs: one to count join dates greater than or
equal to the start date in cell H4:

joined,">=" & start // greater than or equal to start

and one to count join dates less than or equal to the end date in cell H5:

joined,"<=" & end // less than or equal to end

Note that we are concatenating the operators inside the formula with the ampersand (&). The
COUNTIFS function belongs to a group of functions that use this syntax.

With this configuration, COUNTIFS returns the total records with a join date greater than or equal
to the start and end dates in H4 and H5.

Count between age range

The formula above counts records using the start and end dates, but does not take into account
age range. To further restrict the count to the age ranges shown in column G, we need to add two
more range/criteria pairs. The first pair restricts the count to ages greater than or equal to the
"low" number:

age, ">=" & LEFT(G8,FIND("-",G8) - 1) // low

Here, we use the FIND and LEFT function to extract the low number. The FIND function returns
the position of the hyphen (-) and feeds this number (minus 1) to the LEFT function as the
number of characters to extract. LEFT returns zero ("0"), which is concatenated to the greater
than or equal to operator (>=). In the end, we have:

age,">=0"

The second range/criteria pair restricts the count to ages less than or equal to the "high" number in
the age range:

age,"<=" & RIGHT(G8,LEN(G8) - FIND("-",G8)) // high

As before, we use the FIND function to locate the position of the hyphen (-). The result is
subtracted from the total of all characters in the cell (calculated with the LEN function) and this
result is given to the RIGHT function for the number of characters to extract from the right side.
RIGHT returns 20, which is concatenated to the less than or equal to operator (<=). In the end, we
have:

age,"<=20"

As this formula is copied down the range H8:H11, the high and low values in the age ranges are
extracted and used as conditions to restrict the count , while the original date logic remains
unchanged.

Related formulas 

Count cells between dates
The COUNTIFS function is built to count cells that meet multiple
criteria. In this case, we need to provide two criteria: one criteria
for the earlier date and one for the later date. We supply the
named range dates (C5:C11) for both criteria. To...

Count cells between two numbers
The goal in this example is to count numbers that fall within
specific ranges as shown. The lower value comes from the "Start"
column, and the upper value comes from the "End" column. For
each range, we want to include the lower value, but exclude...

Highlight values between
When you use a formula to apply conditional formatting, the
formula is evaluated for each cell in the range, relative to the
active cell in the selection at the time the rule is created. So, in
this case, if you apply the rule to B4:G11, with B4 as...

Highlight dates between
The AND function takes multiple arguments and returns TRUE
only when all arguments return TRUE. The DATE function creates
a proper Excel date with given year, month, and day values.
Because the reference to B4 is fully relative, it will update as...

Summary count by month with COUNTIFS
In this example, we have a list of 100 issues in Columns B to D.
Each issue has a date and priority. We are also using the named
range "dates" for C5:C104 and "priorities" for D5:D105. Starting
in column F, we have a summary table that shows a total...

Related functions 

Excel COUNTIFS Function
The Excel COUNTIFS function returns the count of cells that
meet one or more criteria. COUNTIFS can be used with criteria
based on dates, numbers, text, and other conditions. COUNTIFS
supports logical operators (>,...

Excel FIND Function
The Excel FIND function returns the position (as a number) of
one text string inside another. When the text is not found, FIND
returns a #VALUE error.

Excel LEFT Function
The Excel LEFT function extracts a given number of characters
from the left side of a supplied text string. For example,
LEFT("apple",3) returns "app".

Excel RIGHT Function
The Excel RIGHT function extracts a given number of characters
from the right side of a supplied text string. For example,
RIGHT("apple",3) returns "ple".

Related videos 

How to use the COUNTIFS function
In this video we'll look at how to use the COUNTIFS function to
count cells that meet multiple criteria in a set of data.   

See also 

Excel's RACON functions

Related courses 

Core Formula

Excel Formula Training
Formulas are the key to getting things done in
Excel. In this accelerated training, you'll learn how to
use formulas to manipulate text, work with dates and
times, lookup values with VLOOKUP and INDEX &
MATCH, count and sum with criteria, dynamically
rank values, and create dynamic ranges. You'll also
learn how to troubleshoot, trace errors, and fix
problems. Instant access. See details here.

Key functions

IF function
VLOOKUP function
XLOOKUP function
FILTER function
SUMIFS function
COUNTIFS function
SUMPRODUCT function
INDEX and MATCH
More functions...

Hi - I'm Dave Bruns, and
I run Exceljet with my
wife, Lisa. Our goal is to
help you work faster in

Excel. We create short videos, and clear
examples of formulas, functions, pivot
tables, conditional formatting, and
charts. Read more.

Thanks for this site. It really is quick and to the point. Each article gives the information you need
to solve the problem without excessive descriptions that you usually have to skim through. Keep up
the great work! -Pernilla

